Device

/*
 * Uart57 for TX/RX 
 * Configure no hardware flow control. 
 */
Serial <- hardware.uart57;
Serial.configure(115200, 8, PARITY_NONE, 1, NO_CTSRTS);

local NOCHAR = -1;

/* Log information about this imp module */
server.log("Device started, impee_id " + hardware.getimpeeid() + " and mac = " + imp.getmacaddress() );

/* Pin9 will control an LED to blink, initialize to ground */
led <- hardware.pin9;
led.configure(DIGITAL_OUT, 0);

/* Create a global variable to store current state of 'led' */
state <- 0;


function blink() {
    /* Invert the value of state: */
    state = 1 - state;  

    /* Write current state to 'led' (which is pin9) */
    led.write(state);
    
    /* Clean out any previous characters in any buffers */
    Serial.flush();

    /* Ping the Arduino with the ! character to get the latest data */
    Serial.write("#");
    
    /* Log character received */
    server.log("Data Requested");
    
    /* Wait for initial character to come in */
    local Counter = 0;
    local Result = NOCHAR;
    local InStream = "";
    while ((Result != '\n') || (Result == NOCHAR)) // Keep reading until we see a newline
    {
        Counter = 0;
        while(Result == NOCHAR)
        {
            Result = Serial.read();
    
            if(Result == NOCHAR)
            {
                imp.sleep(0.01);
                if(Counter++ > 100) 
                {
                    server.log("Serial timeout error");
                    return(0);
                }
            }
        }
        
        InStream += format("%c", Result);

        Result = Serial.read(); 
    }
    
    server.log("Received: " + format("%s", InStream)); 
    server.log("STM32 read complete");
    
    // Send info to agent, that will in turn push to internet
    agent.send("postToWU", InStream);
    
    /* Schedule the imp to wakeup in 0.5 seconds and call blink() again  */
    imp.wakeup(60, blink);
}

/* Start the loop */
blink();

Agent

local STATION_ID = "KMILINDE6";
local STATION_PW = "jpmjfdzd"

// When we hear something from the device, split it apart and post it
device.on("postToWU", function(dataString) {
    
    //server.log("Incoming: " + dataString);
    
    //Break the incoming string into pieces by comma
    a <- mysplit(dataString,',');

    if(a[0] != "#" || a[3] != "#")
    {
        server.log(format("Error: incorrect frame received (%s, %s)", a[0], a[3]));
        server.log(format("Received: %s)", dataString));
        return(0);
    }
    
    //a[0] is #
    local humidity = a[1];
    local tempf = a[2];
    //a[3] is #
    
    currentTemp <- mysplit(tempf, '=');

    //Correct for humidity out of bounds
    currentHumidity <- mysplit(humidity, '=');
    if(currentHumidity[1].tointeger() > 99) humidity = "humidity=99";
    if(currentHumidity[1].tointeger() < 0) humidity = "humidity=0";


    //Now we form the large string to pass to wunderground
    local strMainSite = "http://rtupdate.wunderground.com/weatherstation/updateweatherstation.php";
    local strID = "ID=" + STATION_ID;
    local strPW = "PASSWORD=" + STATION_PW;

    //Form the current date/time
    //Note: .month is 0 to 11!
    local currentTime = date(time(), 'u');
    local strCT = "dateutc=";
    strCT += currentTime.year + "-" + format("%02d", currentTime.month + 1) + "-" + format("%02d", currentTime.day);
    strCT += "+" + format("%02d", currentTime.hour) + "%3A" + format("%02d", currentTime.min) + "%3A" + format("%02d", currentTime.sec);

    local WUString = strMainSite;
    WUString += "?" + strID;
    WUString += "&" + strPW;
    WUString += "&" + strCT;
    WUString += "&" + humidity;
    WUString += "&" + tempf;
    WUString += "&" + "softwaretype=BeningoWeatherImp"; //Cause we can
    WUString += "&" + "realtime=1"; //You better believe it!
    WUString += "&" + "rtfreq=3"; //Set rapid fire freq to once every 10 seconds
    WUString += "&" + "action=updateraw";

    server.log("string to send: " + WUString);
    
    //Push to Wunderground
    local request = http.post(WUString, {}, "");
    local response = request.sendsync();
    server.log("Wunderground response = " + response.body);
    server.log("Update complete!");
}); 

//From Hugo: http://forums.electricimp.com/discussion/915/processing-nmea-0183-gps-strings/p1
//You rock! Thanks Hugo!
function mysplit(a, b) {
  local ret = [];
  local field = "";
  foreach(c in a) {
      if (c == b) {
          // found separator, push field
          ret.push(field);
          field="";
      } else {
          field += c.tochar(); // append to field
      }
   }
   // Push the last field
   ret.push(field);
   return ret;

